# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: Jeff Mickey <jeff@archlinux.org>

pkgname=castget
pkgver=2.0.1
pkgrel=1
pkgdesc="Command-line based RSS enclosure downloader primarily intended for downloading of podcasts"
arch=('loong64' 'x86_64')
url="https://johndal.com/castget/"
license=('GPL')
depends=('glib2' 'libxml2' 'curl' 'id3lib')
changelog=$pkgname.changelog
source=(https://download.savannah.gnu.org/releases/$pkgname/$pkgname-$pkgver.tar.bz2{,.sig})
validpgpkeys=('561BCBC3DF8A53503A62416EA7DE43E1D881FF60') # Marius Larsen Jøhndal <mariuslj@ifi.uio.no>
sha256sums=('438b5f7ec7e31a45ed3756630fe447f42015acda53ec09202f48628726b5e875'
            'SKIP')

build() {
  cd "${srcdir}"/$pkgname-$pkgver

  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}"/$pkgname-$pkgver

  make DESTDIR="${pkgdir}" install
}
